

```

/** ----- 10 RIVERS -----
/** FILE NAME:          10_Rivers.txt
/** AUTHOR:            RENEE SCHICKER
/** SCRIPT CREATED:    19 JANUARY 2009
/** UPDATED/MODIFIED:  30 SEPTEMBER 2009
/**
/** The scripts may be supplied in a more readily useable format if the work is acknowledged
/** CONTACT:          Renee_Schicker@hotmail.com
/**
/** SCRIPT USED BY:      00_MASTER.txt
/** USES THE SCRIPT:    CheckProgEdit.txt      (CREATED 2 MARCH 2009)
/**
/** INPUT COVERS:
/**     river_cl        D:\renee_GIS\input_data\river_cl
/**     DEM_Bnd         D:\Renee_GIS\Output_data\Organised\03_DEM\DEM_Bnd
/**
/** OUTPUT COVERS:      region_rivers          Rivers_Dist          RiversDist2
/** OUTPUT GRID:        Rivergrid
/**
/** TEMP. COVERS:       rivbuff%dist_1%        rivbuff%dist_2%
/**                     rivbuff%dist_3%        rivbuff%dist_4%
/**                     rivb%dist_3%_%dist_4%   rivb%dist_2%_%dist_4%
/**                     rivers_buff
/**
/** FUNCTIONS USED:     &CALL                   &RETURN                   &ROUTINE
/**                     &IF &THEN               [EXIST]                   KILL
/**                     BUFFER                   ADDITEM                   &RUN
/**                     EDITCOVER (EC)           EDITFEATURE (EF)         SELECT
/**                     CALCULATE                 SAVE                       QUIT (Q)
/**                     UNION                     DROPITEM                 &SETVAR
/**                     CLIP                      POLYGRID
/**
/** PURPOSE:           Clip river classification layer to region using DEM_Bnd (to match DEM
/**                     boundary). Buffer rivers to create polygon features which will mark a set
/**                     distance range and classify these accordingly by adding a new field and
/**                     making the respective edits. All buffer layers are unioned so the largest is
/**                     first in line and each subsequently smaller buffer zone unioned over top. The
/**                     second lot of edits are made in the same way so that the smaller distances
/**                     won't be misclassified. All fields brought in through the union that are no
/**                     longer required are removed using dropitem command. The combined
/**                     buffer layer is clipped to the DEM_Bnd as the buffer zones extend beyond
/**                     the region as a result of the process. This should then be rasterised.
/**
/** OTHER NOTES:       May need to decide on appropriate distances at a later date - find in
/**                     Literature & work out how to apply to area.
/**
/** ----- HISTORY -----
/** 15 DECEMBER 2008    3_clip.txt individual clip script created specifically to be run from a
/**                     master script as part of the processing of parameters for Waikato
/**                     region. Important to check input exists before processing.
/** 19 JANUARY 2009     Created 11_river_buff.txt

```

```

/**      12 FEBRUARY 2009      Can now clip with the digitised regional boundary I made (which has
/**                               a more representative coast line than the EW one) based on the
/**                               district boundaries layer and cut off points based on the EW layer
/**      02 MARCH 2009      Added &RUN CheckProgEdit.txt process
/**      06 MARCH 2009      Moved to new computer - filepaths changed
/**      9-10 MARCH 2009     Formatted listed everything and added descriptions and history.
/**      11 MARCH 2009      ADDED variables (&SETVAR PROCESS) so you only have to
/**                               change the value of the distances once (at start of script)
/**                               as opposed to everywhere they appear in the script.
/**                               Now runs from 00_MASTER.txt instead of 0_parent.txt. Have
/**                               changed file name to 10_river_buff.txt from 11_river_buff.txt.
/**      27 MARCH 2009      3_Clip.txt renamed 04_Clip.txt. Check script is consistent with
/**                               others, update script information. Added CLIP rivers_buff by
/**                               region_bnd then KILL rivers_buff afterwards.
/**      30 APRIL 2009      First attempt to rasterise vector data using POLYGRID
/**      31 APRIL-05 MAY 2009 Rasterising rivers resulted in an error, probably too much info to
/**                               take on??
/**      20 MAY 2009      Separate Input data and output data directories, so workspace is set
/**                               to a separate output folder, so reduces the chance of deleting input
/**                               data by accident.
/**      11 JUNE 2009      Incorporated the clip rivers process from 04_clip.txt and rasterise
/**                               command based on idea from 15_Rasterise.txt, although it probably
/**                               won't work as I haven't dealt with the resulting error yet.
/**                               Have renamed 10_river_buff.txt as 12_Rivers.txt.
/**                               May need to adjust the distance variables, might be a bit too small?
/**      21 JUNE 2009      Added the All_areas Routine to union the buffered rivers to the
/**                               DEM_Bnd then can assign 0 for all areas outside the buffer region
/**                               (to match the values of any trapped polygons) which should reduced
/**                               the NO DATA (-9999) values when converted to Grid.
/**      30 SEPTEMBER 2009   Added separate workspaces for each script, so have to add file path
/**                               to find input files, also corrected input and output sections.
/** *****
/** *****

```

```

/* Set the distance variables used in the buffering

```

```

&SETVAR dist_1 = 100          /* Smallest distance
&SETVAR dist_2 = 200
&SETVAR dist_3 = 300
&SETVAR dist_4 = 400          /* Greatest distance

```

```

&CALL Clip_1
&CALL buffer
&CALL Add_Edit                /* USES: CheckProgEdit.txt
&CALL union
&CALL kill
&CALL Edit_Drop              /* USES: CheckProgEdit.txt
&CALL End_Clip
&CALL ALL_AREAS
&CALL Rasterise

```

&RETURN

```
/*  
*****  
*****  
*****
```

&ROUTINE Clip_1

```
&IF [EXIST region_rivers -cover] &THEN KILL region_rivers  
CLIP D:\renee_gis\input_data\river_cl D:\Renee_GIS\Output_data\Organised\03_DEM\DEM_Bnd  
region_rivers LINE 1 /* Clip RIVERS  
&TYPE rivers clipped  
&RETURN  
/*  
*****
```

&ROUTINE buffer

```
&type RIVER buffer time!  
BUFFER region_rivers rivbuff%dist_1% # # %dist_1% 1 LINE ROUND  
BUFFER region_rivers rivbuff%dist_2% # # %dist_2% 1 LINE ROUND  
BUFFER region_rivers rivbuff%dist_3% # # %dist_3% 1 LINE ROUND  
BUFFER region_rivers rivbuff%dist_4% # # %dist_4% 1 LINE ROUND  
&RETURN  
/*  
*****
```

&ROUTINE Add_Edit

```
/* add string class for id  
/* additem <in_info> <out_info> <item_name> <item_width> <output_width> <item_type>
```

```
&type add a column to each buffer layer...  
ADDITEM rivbuff%dist_1%.pat rivbuff%dist_1%.pat buff_dist1 5 5 I  
ADDITEM rivbuff%dist_2%.pat rivbuff%dist_2%.pat buff_dist2 5 5 I  
ADDITEM rivbuff%dist_3%.pat rivbuff%dist_3%.pat buff_dist3 5 5 I  
ADDITEM rivbuff%dist_4%.pat rivbuff%dist_4%.pat buff_dist4 5 5 I
```

```
/* ---- Make Edits-----  
/* Need to use ArcEdit so run the associated script to do this  
/*&RUN CheckProgEdit.txt  
&RUN d:\renee_gis\scripts\CheckProgEdit.txt
```

```
&TYPE entering buffer distances into new individual buff_dist attribute columns  
/* enter buffer distances
```

```
EC rivbuff%dist_1%  
EF polygon  
SELECT for INSIDE = 100  
CALCULATE buff_dist1 = %dist_1%  
SAVE
```

```
EC rivbuff%dist_2%  
EF polygon
```

```
SELECT for INSIDE = 100
CALCULATE buff_dist2 = %dist_2%
SAVE
```

```
EC rivbuff%dist_3%
EF polygon
SELECT for INSIDE = 100
CALCULATE buff_dist3 = %dist_3%
SAVE
```

```
EC rivbuff%dist_4%
EF polygon
SELECT for INSIDE = 100
CALCULATE buff_dist4 = %dist_4%
SAVE
```

```
Q
```

```
&RETURN
```

```
/******
```

```
&ROUTINE union
```

```
/*UNION <in_cover> <union_cover> <out_cover> {fuzzy_tolerance} {JOIN | NOJOIN}
```

```
&IF [EXIST rivers_buff -COVER] &THEN KILL rivers_buff ALL
```

```
&TYPE multi stage union begins...
```

```
UNION rivbuff%dist_3% rivbuff%dist_4% rivb%dist_3_%dist_4%
```

```
UNION rivbuff%dist_2% rivb%dist_3_%dist_4% rivb%dist_2_%dist_4%
```

```
UNION rivbuff%dist_1% rivb%dist_2_%dist_4% rivers_buff
```

```
&RETURN
```

```
/******
```

```
&ROUTINE kill
```

```
&IF [EXIST rivbuff%dist_1% -COVER] &THEN KILL rivbuff%dist_1% ALL
```

```
&IF [EXIST rivbuff%dist_2% -COVER] &THEN KILL rivbuff%dist_2% ALL
```

```
&IF [EXIST rivbuff%dist_3% -COVER] &THEN KILL rivbuff%dist_3% ALL
```

```
&IF [EXIST rivbuff%dist_4% -COVER] &THEN KILL rivbuff%dist_4% ALL
```

```
&IF [EXIST rivb%dist_3_%dist_4% -COVER] &THEN KILL rivb%dist_3_%dist_4% ALL
```

```
&IF [EXIST rivb%dist_2_%dist_4% -COVER] &THEN KILL rivb%dist_2_%dist_4% ALL
```

```
&RETURN
```

```
/******
```

```
&ROUTINE Edit_Drop
```

```
ADDITEM rivers_buff.pat rivers_buff.pat Riverdist 5 5 I
```

```
/* Need to use ArcEdit so run the associated script to do this
```

```
/*&RUN CheckProgEdit.txt
```

```
&RUN d:\renee_gis\scripts\CheckProgEdit.txt
```

&TYPE entering all buffer distances into new Riverdist attribute column
/* enter buffer distances

EC rivers_buff
EF polygon

SELECT for buff_dist4 = %dist_4%
CALCULATE Riverdist = %dist_4%

SELECT for buff_dist3 = %dist_3%
CALCULATE Riverdist = %dist_3%

SELECT for buff_dist2 = %dist_2%
CALCULATE Riverdist = %dist_2%

SELECT for buff_dist1 = %dist_1%
CALCULATE Riverdist = %dist_1%
SAVE
Q

/* delete unneeded variables since they've served their purpose
/* delete temporary attribute columns
&TYPE goodbye temporary buff_dist variables

DROPITEM rivers_buff.pat rivers_buff.pat BUFF_DIST1 BUFF_DIST2 BUFF_DIST3 BUFF_DIST4
DROPITEM rivers_buff.pat rivers_buff.pat RIVBUFF%dist_1%# RIVBUFF%dist_1%-ID
DROPITEM rivers_buff.pat rivers_buff.pat RIVBUFF%dist_2%# RIVBUFF%dist_2%-ID
DROPITEM rivers_buff.pat rivers_buff.pat RIVBUFF%dist_3%# RIVBUFF%dist_3%-ID
DROPITEM rivers_buff.pat rivers_buff.pat RIVBUFF%dist_4%# RIVBUFF%dist_4%-ID
DROPITEM rivers_buff.pat rivers_buff.pat RIVB%dist_3%_%dist_4%# RIVB%dist_3%_%dist_4%-ID
DROPITEM rivers_buff.pat rivers_buff.pat RIVB%dist_2%_%dist_4%# RIVB%dist_2%_%dist_4%-ID
&RETURN
/*****

&ROUTINE End_Clip

&IF [EXIST Rivers_Dist -COVER] &THEN KILL Rivers_Dist ALL
CLIP rivers_buff D:\Renee_GIS\Output_data\Organised\03_DEM\DEM_Bnd Rivers_Dist POLY 1
&IF [EXIST rivers_buff -COVER] &THEN KILL rivers_buff ALL
&RETURN
/*****

&ROUTINE ALL_AREAS
/** Cover extent of the region so don't have missing pieces.

&IF [EXIST Rivers_Dist2 -COVER] &THEN KILL Rivers_Dist2 ALL
UNION Rivers_Dist D:\Renee_GIS\Output_data\Organised\03_DEM\DEM_Bnd Rivers_Dist2

ADDITEM Rivers_Dist2.pat Rivers_Dist2.pat River_dist 5 5 I

/* Need to use ArcEdit so run the associated script to do this

&RUN d:\renee_gis\scripts\CheckProgEdit.txt

&TYPE entering all buffer distances into new Road_dist attribute column
/* enter buffer distances

EC Rivers_Dist2
EF polygon

SELECT ALL
CALCULATE River_dist = -9999

SELECT for GRID-CODE = 1
CALCULATE River_dist = 0

SELECT for Riverdist = %dist_4%
CALCULATE River_dist = %dist_4%

SELECT for Riverdist = %dist_3%
CALCULATE River_dist = %dist_3%

SELECT for Riverdist = %dist_2%
CALCULATE River_dist = %dist_2%

SELECT for Riverdist = %dist_1%
CALCULATE River_dist = %dist_1%

SAVE
Q

DROPITEM Rivers_Dist2.pat Rivers_Dist2.pat DEM_BND# DEM_BND-ID GRID-CODE
DROPITEM Rivers_Dist2.pat Rivers_Dist2.pat Rivers_Dist# Rivers_Dist-ID Riverdist
&RETURN

/*****
&ROUTINE Rasterise

&IF [EXIST Rivergrid -GRID] &THEN KILL Rivergrid ALL
POLYGRID Rivers_Dist2 Rivergrid River_dist

25
Y